
ABSTRACT  

Love it or hate it, Microsoft Excel is used extensively throughout the business world. As a SAS user, you 
can enhance the impact of your work by using the ODS EXCEL destination to create high-quality, 
customized output in Excel format directly from SAS. This paper walks through a series of examples 
demonstrating the flexibility and power of this approach. In addition to complete control over visual 
attributes such as fonts, colors, and borders, the ODS EXCEL destination allows the SAS user to take 
advantage of Excel features such as multiple tabs, frozen or hidden rows and columns, and even Excel 
formulas to deliver the high-impact results you and your customers want! 

OVERVIEW OF THE OUTPUT DELIVERY SYSTEM  

Before SAS introduced the Output Delivery System, the traditional SAS listing was the primary 
mechanism by which the SAS system displayed output to the user.  While this arrangement served its 
purpose, it had several limitations.  Since it was designed at a time when line printers were widely used, it 
used a monospace typeface in which characters could be neatly arranged in rows and columns.  It did not 
allow for the use of colors, fonts, graphics, and other visual elements commonly used today.  The listing 
output could also be difficult to import into other applications. 

To overcome these limitations and facilitate the creation of publication-quality output, SAS developed the 
Output Delivery System (ODS).  It has been part of the base SAS product since version 7.  No separate 
license is required to access this functionality. 

ODS provides extensive flexibility for customizing SAS output: 

• Output can be created in a variety of common file formats (including Microsoft Excel files). 

• Colors, fonts, borders, shading, and other visual elements can be incorporated. 

• Graphics, images, links, and tables can be embedded. 

• Pagination can be precisely controlled. 

• Accessible output can be generated that will work with assistive technology. 

In order to allow output to be created in many different file formats, ODS introduces the concept of a 
“destination.”  An ODS destination is simply a representation of a location and format to which output can 
be routed.  Some commonly used ODS destinations include RTF, HTML, PDF, POWERPOINT, and 
EXCEL.  Traditional SAS listing output is also still available through the LISTING destination.  This paper 
will focus exclusively on the use of the ODS EXCEL destination. 

USING THE ODS EXCEL DESTINATION 

The ODS EXCEL destination creates native XLSX files, just like the files created by modern versions of 
Microsoft Excel.  To create such a file using SAS, simply include a pair of ODS EXCEL statements 
(sometimes referred to as the “ODS sandwich”) around the SAS code which generates the desired 
output.  The first ODS EXCEL statement opens the EXCEL destination and directs the output to a specific 
file location using the FILE= option, while the second statement closes the destination using the CLOSE 
option.  For example, the following sequence of statements directs the output from the PRINT procedure 
to the external Microsoft Excel file shown in Figure 1. 

ods excel file="c:\example.xlsx";  

proc print data=sashelp.cars; run; 

ods excel close; 



2 

 
Figure 1. ODS Excel output from the PRINT Procedure 

THE ODS EXCEL OPTIONS OPTION 

The ODS EXCEL statement has an OPTIONS option that allows the user to specify a variety of Excel-
specific sub-options.  One or more pairs of sub-options and values can be enclosed in parentheses after 
the OPTIONS keyword as follows: 

ods excel file="c:\example.xlsx" 

      options(suboption1=value1 suboption2=value2); 

 

The next several examples demonstrate the use of some of the sub-options available using the OPTIONS 
option.  An exhaustive list of the available sub-options can be found in SAS® 9.4 Output Delivery System: 
User’s Guide, Fifth Edition (2016). 

SPECIFYING THE WORKSHEET NAME 

An Excel workbook consists of one or more worksheets, each of which has a name.  By default, the 
worksheets in an Excel workbook created using the ODS EXCEL destination will be named automatically 
by SAS.  However, if a particular worksheet name is desired, it can be specified using the SHEET_NAME 
sub-option of the OPTIONS option.   

The following code prints the SASHELP.CARS dataset to an Excel worksheet named “My CARS Output” 
as displayed in Figure 2. 

ods excel file="c:\example.xlsx" 

      options(sheet_name="My CARS Output");  

proc print data=sashelp.cars; run; 

ods excel close; 

Figure 2. Specifying the Worksheet Name 

CREATING MULTIPLE WORKSHEETS WITH BY GROUPS 

Many SAS procedures support BY group processing using the BY statement.  This is conceptually similar 
to running the procedure separately for each BY group.  When BY group processing is used in 
conjunction with the ODS EXCEL destination, a separate worksheet is created within the Excel workbook 
for each unique value of the BY variable (or each unique combination of values when there are multiple 
BY variables). 

The naming of the individual worksheet tabs generated during BY group processing can be controlled 
using the #BYVALn syntax within the SHEET_NAME sub-option described in the previous example.  For 
example, if #BYVAL1 were used, it would be replaced in each worksheet name with the value of the first 
BY variable.  Similarly, #BYVAL2 would be replaced with the value of the second BY variable, if any. 

Of course, in order to use BY group processing, the input dataset must be sorted (or indexed) by the BY 
variable(s).  The following example uses the SORT procedure to sort the data before opening the ODS 
EXCEL destination and invoking PROC PRINT to create the output seen in Figure 3. 

 

proc sort data=sashelp.cars out=cars; 

  by type make model; 

run; 

 

ods excel file="c:\example.xlsx" 

      options(sheet_name="#byval1");  

proc print data=cars; 

   by type; 

run; 

ods excel close; 

Figure 3. Multiple Worksheets with BY Groups 

 

The value(s) of the BY variable(s) will also appear at the top of each worksheet as a header line in the 
first row of the spreadsheet. To suppress this header, invoke the NOBYLINE option on the global 
OPTIONS statement (not to be confused with the OPTIONS option on the ODS EXCEL statement). 

options nobyline; 

SPECIFYING A START LOCATION 

By default, any SAS output created using the ODS EXCEL destination will begin in the upper-left corner 
of the Excel worksheet, which Excel designates as cell A1.  To place the output in a different location in 
the worksheet, use the START_AT sub-option in the OPTIONS option to specify a different cell.  The 
following code would start the output in cell C3 instead. Output is shown in Figure 4. 

ods excel file="c:\example.xlsx" options(start_at="C3");  

proc print data=sashelp.cars; run; 

ods excel close; 

 

 

Figure 4. Specifying a Start Location 

ADJUSTING COLUMN WIDTHS AND ROW HEIGHTS 

ODS automatically determines the width for each column in the Excel worksheet based on the lengths of 
the values and/or labels being placed in that column.  You can manually specify the column widths in a 
couple of different ways.  If you wish to specify different widths for different columns, this can be 
accomplished using in-line styling, which is discussed later in this paper.   

However, if you simply wish to specify a common width for all columns in the workbook, this can be done 
with the ABSOLUTE_COLUMN_WIDTH sub-option on the OPTIONS option.  The width can be specified 
using any ODS-supported unit (cm, mm, in, px, etc.).  If you omit the units, the width value will translate 
directly into Excel’s default width units which represent a specific number of characters in the default font. 

The following ODS EXCEL statement sets the column width to 3 inches. Output is shown in Figure 5. 

ods excel file="c:\example.xlsx" 

      options(absolute_column_width="3in"); 

proc print data=sashelp.cars; run; 

ods excel close; 

 

 

Figure 5. Adjusting Absolute Column Width 

Similarly, to change the height of all rows in an Excel worksheet, use the ABSOLUTE_ROW_HEIGHT 
sub-option. Output is shown in Figure 6. 

ods excel file="c:\example.xlsx" 

      options(absolute_row_height="30"); 

proc print data=sashelp.cars; run; 

ods excel close; 

 

 

Figure 6. Adjusting Absolute Row Height 

If you wish to specify a height for the header row that is different from the rest of the rows, use the 
ROW_HEIGHTS sub-option and provide the header row height followed by a comma and then the body 
row height. Figure 7 displays the result. 

ods excel file="c:\example.xlsx" 

      options(row_heights="30,15"); 

proc print data=sashelp.cars; run; 

ods excel close; 

Figure 7. Adjusting Row Heights 

HIDING OR FREEZING ROWS AND COLUMNS 

The extensive functionality provided by Microsoft Excel includes the ability to hide individual or multiple 
rows and/or columns as well as to freeze header rows and/or columns.  Hidden rows or columns still exist 
in the spreadsheet but simply are not visible until unhidden.  Frozen header rows or columns are locked 
in place and always remain visible as the rest of the spreadsheet is scrolled in any direction.   

Both features can be accessed using the ODS EXCEL statement.  To hide specific rows and/or columns, 
use the HIDDEN_ROWS and/or HIDDEN_COLUMNS sub-options to specify the row or column number of 
the row or column to be hidden.  To hide multiple rows or columns, list all the numbers separated by 
commas.  To freeze header rows or columns, use the FROZEN_ROWHEADERS and/or 
FROZEN_HEADERS sub-options to specify the number of rows and/or columns to be frozen.  For 
example, if the FROZEN_HEADERS sub-option is given a value of “1” then only the top row will be frozen 
whereas a value of “3” would result in the first three rows all being frozen. 

The following code creates an Excel worksheet having the first two rows and the first column (column A) 
frozen.  The resulting output in Figure 8 has been scrolled both down and to the right but the first two 
rows and the first column always remain visible. 

ods excel file="c:\example.xlsx" 

 options( 

  frozen_headers="1" 

  frozen_rowheaders="2"); 

proc print data=sashelp.cars; run; 

ods excel close; 

 

 

Figure 8. Frozen Headers and Row Headers 

Similarly, the following code creates the Excel worksheet shown in Figure 9 having the 5th and 7th rows 
and 6th and 8th columns (that is, columns F and H) hidden. 

ods excel file="c:\example.xlsx" 

 options( 

  hidden_rows="5,7" 

  hidden_columns="6,8"); 

proc print data=sashelp.cars; run; 

ods excel close; 

 

 
Figure 9. Hidden Row and Columns 

REMOVING OR MODIFYING CELL BORDERS 

Another feature provided by the ODS EXCEL destination is the ability to remove or modify the Excel 
worksheet cell borders.  Unfortunately, this is not available as a sub-option on the OPTIONS option.  
Rather, we must create a custom style using the TEMPLATE procedure and then apply that style on the 
ODS EXCEL statement.  Custom styles are discussed in more detail later in this paper. 

To remove the cell borders from the worksheet, use the following code to create a custom style which 
derives from the built-in style named STYLES.EXCEL but has the BORDERSTYLE attribute set to 
HIDDEN.  The classes DATA, HEADER, and ROWHEADER are included so that this style modification 
will be applied to all areas of the output as shown in Figure 10. 

Note the use of the ODS PATH statement to ensure that this custom style can be found when it is applied 
using the STYLE= option on the ODS EXCEL statement. 

ods path(prepend) work.templat(update); 

proc template; 

  define style styles.mystyle; 

    parent=styles.excel; 

    class data, header, rowheader / borderstyle=hidden; 

  end; 

run; 

ods excel file="c:\example.xlsx" style=styles.mystyle; 

proc print data=sashelp.cars; run; 

ods excel close; 

Figure 10. Removing the Cell Borders 

As an alternative, to modify the appearance of the cell borders, change the values of the 
BORDERSTYLE, BORDERWIDTH, and/or BORDERCOLOR attributes as desired.  The example below 
creates the Excel worksheet displayed in Figure 11 with solid red cell borders that are 3 pixels in 
thickness.   

ods path(prepend) work.templat(update); 

proc template; 

  define style styles.mystyle; 

    parent=styles.excel; 

    class data, header, rowheader / 

      borderstyle=solid 

      borderwidth=3px 

      bordercolor=red; 

  end; 

run; 

ods excel file="c:\example.xlsx" style=styles.mystyle; 

proc print data=sashelp.cars; run; 

ods excel close; 

 

Figure 11. Modifying the Cell Borders 

CREATING EXCEL FORMULAS FROM SAS 

Formulas are one of the most powerful features of Excel, and the ODS EXCEL destination provides a 
way to build Excel formulas directly from SAS.  In general, ODS EXCEL will format any character data as 
a text string in Excel.  However, if the FORMULAS sub-option of the OPTIONS option is set to “YES” then 
any string beginning with an equal sign will be interpreted as an Excel formula. 

To make use of this functionality, it is incumbent upon the programmer to utilize the tools available within 
SAS to construct the desired Excel formula.  In the following example, the SASHELP.CARS dataset is 
pre-processed in a DATA step to add a new character variable called PRICEDIFF.  This variable contains 
an Excel formula constructed using the CATS function.  The formula computes the difference between 
the variables MSRP (found in column C of the Excel output) and INVOICE (found in column D).  

data cars; 

  set sashelp.cars; 

  pricediff = cats("=C",_n_+1,"-D",_n_+1); 

run; 

 

Next, we use the PRINT procedure on the resulting WORK.CARS dataset, directing the output to the 
ODS EXCEL destination.  The FORMULAS sub-option is set to “YES” so that the text found in 
PRICEDIFF will be interpreted as Excel formulas.  Note that some additional in-line styling is applied to 
PRICEDIFF so that it will be right-justified and will have the appropriate Excel formatting applied so that it 
will be rendered as a dollar amount.  

ods excel file="c:\example.xlsx" options(formulas="YES"); 

proc print data=cars noobs label; 

  label pricediff="Price Difference"; 

  var make model msrp invoice ; 

  var pricediff / style={just=r tagattr='format:$#,##0_)'}; 

run; 

ods excel close; 

The output is shown in Figure 12 with cell E2 selected.  The formula bar at the top shows that this cell 
contains a live Excel formula, not a static value. 

 

Figure 12. Creating Excel Formulas from SAS 

OUTPUT FROM MULTIPLE PROCEDURES 

Earlier we discussed how to create a separate worksheet (or tab) within an Excel workbook (file) for each 
BY group within the output from a single procedure.  We can also create multiple worksheets by running 
multiple procedures.  By default, output from each SAS procedure is written to a separate worksheet.  We 
can individually customize the attributes of each worksheet using additional ODS EXCEL statements.  
The following example creates two separate worksheets, each with a specific color and name as shown 
in Figure 13. 

ods excel file="c:\msug\example.xlsx" 

  options(tab_color="red" sheet_name="Cars"); 

  

proc print data=sashelp.cars; run; 

 

ods excel options(tab_color="green" sheet_name="Class"); 

 

proc print data=sashelp.class; run; 

 

ods excel close; 

 

 

Figure 13. Output from Multiple Procedures 

MULTIPLE PROCEDURES WITH BY GROUPS 

Running multiple procedures using BY group processing results by default in a separate worksheet for 
each BY group for each procedure.  For example, running three procedures using BY group processing 
with six BY groups would result in an Excel workbook having 18 worksheets. 

If this is not desirable, it is possible to create an Excel workbook in which all the procedure output for a 
given BY group is placed together in a single worksheet so that the workbook contains just one worksheet 
per BY group.  In the following example, we accomplish this by taking manual control of BY group 
processing using the SAS Macro Language. 

The code is placed within a macro definition so that can make use of a macro %DO loop.  The first step is 
to use the SQL procedure to create a series of macro variables containing the unique values of the BY 
variable.  In this example, we use the SASHELP.CARS dataset with TYPE as the BY variable.  The 
macro variables are named TYPE1, TYPE2, etc.  We also create one additional macro variable, 
NUMTYPES, containing the number of unique values of TYPE, which happens to be 6. 

%macro create_xlsx; 

 

  proc sql noprint; 

    select distinct(type) into :type1- 

      from sashelp.cars 

      order by type; 

    %let numtypes = &sqlobs; 

  quit; 

Next, we disable the automatic titles generated by each SAS procedure and open the ODS EXCEL 
destination.  The SHEET_INTERVAL sub-option on the OPTIONS option is set to “NONE”.  This causes 
all procedure output to be directed to the same worksheet until we indicate otherwise. 

  ods noproctitle; 

  ods excel file="c:\example.xlsx" 

      options(sheet_interval='none' embedded_titles='on'); 

We then invoke a macro %DO loop which executes once for each unique value of TYPE.  During each 
iteration, the Excel worksheet name is set to match the current value of TYPE.  Next, a single title is 
added to identify the current value of TYPE.  

Then, the MEANS, FREQ, and SGPLOT procedures are called.  However, no BY statement is used.  
Instead, the input dataset is subset using a WHERE statement so that only the records for the current 
value of TYPE are processed.  

Once all three procedures have been run, the ODS EXCEL statement is invoked with the 
SHEET_INTERVAL sub-option set to “NOW” to indicate that the next procedure output should be written 
to a new worksheet. 

  %do i = 1 %to &numtypes; 

    ods excel options(sheet_name="&&type&i");  

    title "Vehicle Type = &&type&i"; 

    proc means data=sashelp.cars;  

      where type = "&&type&i"; 

      var horsepower; 

    run; 

    title; 

    proc freq data=sashelp.cars;  

      where type = "&&type&i"; 

      tables origin; 

    run; 

    proc sgplot data=sashelp.cars;  

      where type = "&&type&i"; 

      reg x=weight y=enginesize; 

    run; 

    ods excel options(sheet_interval='now'); 

  %end; 

Finally, once the %DO loop has completed, the Excel file is closed.  This ends the macro definition.  The 
macro must then be executed to produce the output shown in Figure 14. 

  ods excel close; 

%mend create_xlsx; 

 

%create_xlsx; 

Refer to Horstman (2020) for a more detailed explanation of utilizing sets of macro variables in this 
manner to create dynamic programs that execute code independently for each unique value (or 
combination of values) of a specific variable (or group of variables) without hardcoding those values. 

Figure 14. Multiple Procedures with BY Groups 

WORKING WITH DATES AND TIMES 

SAS and Microsoft Excel use different methods to represent date and time values.  SAS stores dates as 
the number of days since January 1, 1960, while Microsoft Excel instead counts the days since January 
1, 1900.   

Times in SAS are represented by the number of seconds since midnight, while Excel converts the time of 
day into a decimal value representing the fraction of a day that would have passed at that clock time.  In 
other words, Excel stores noon as 0.5 (halfway through the day) and SAS stores noon as 43,200. 

When creating an Excel spreadsheet using the ODS EXCEL destination that includes date and/or time 
values, it’s important to make sure that these values are correctly converted.  To do this, simply be sure 
that an appropriate date or time format is applied in SAS to any variables which are used in your Excel 
output.  Unformatted values will not convert correctly since ODS EXCEL will be unaware of what they 
represent. 

CUSTOMIZING ODS EXCEL OUTPUT WITH STYLES 

The previous section described how to create an Excel workbook directly from SAS using the ODS 
EXCEL destination and how to modify some basic properties and features of the resulting Excel file.  
ODS styles take this one step further by providing rich functionality for the complete customization of 
nearly every aspect of the appearance of your Excel workbook.  This includes colors, fonts, borders, 
backgrounds, and much more.  We will briefly discuss three ways to access this functionality:  built-in 
styles, custom styles, and in-line styling. 

USING BUILT-IN STYLES 

The SAS system includes a library of built-in styles that are ready for use.  Each of these styles 
represents a complete set of visual attributes that can be applied to any ODS output, whether that’s an 
Excel spreadsheet or another ODS destination.  Some of the built-in styles that are included with SAS are 
listed below: 

• STYLES.BANKER 

• STYLES.BRICK 

• STYLES.HARVEST 

• STYLES.OCEAN 

• And many more… 

These built-in styles are stored in the SASHELP.TMPLMST item store.  To see a list of available built-in 
styles, submit the following PROC TEMPLATE code: 

proc template; 

   path sashelp.tmplmst; 

   list styles; 

run; 

To apply one of these styles to Excel output generated through the ODS EXCEL destination, utilize the 
STYLE= option on the ODS EXCEL statement when opening the destination.  The following example 
prints the SASHELP.CARS dataset using the STYLES.BROWN style as shown in Figure 15. 

ods excel file="c:\example.xlsx" style=styles.brown; 

proc print data=sashelp.cars; run; 

ods excel close; 

 

 

Figure 15. Using Built-in Styles 

CREATING YOUR OWN STYLES 

In addition to using the wide variety of built-in styles described above, ODS also allows you to define your 
own custom styles using PROC TEMPLATE.  This gives you the flexibility to specify the values of each of 
the different visual attributes that control every aspect of how your Excel output appears. 

You can either create a new style completely from scratch, or you can start with an existing style and 
modify it as desired.  The following PROC TEMPLATE code creates a new style called 
STYLES.MYSTYLE which inherits the properties of the built-in style STYLES.PEARL and changes the 
value of a single attribute. 

proc template; 

 define style styles.mystyle; 

  parent = styles.pearl; 

  class data / color=red; 

 end; 

run; 

This style can then be applied to ODS EXCEL output as before to produce the result in Figure 16. 

ods excel file="c:\example.xlsx" style=styles.mystyle; 

proc print data=sashelp.cars; run; 

ods excel close; 

 

Figure 16. Creating a User-Defined Style 

PROC TEMPLATE is a complex procedure with an extensive array of available classes, elements, and 
attributes.  Detailed information can be found in the PROC TEMPLATE documentation in SAS 9.4 Output 
Delivery System: Procedures Guide, Third Edition (2016), or refer to Smith (2013) for a thorough 
treatment of the topic. 

USING IN-LINE STYLING 

The final way to employ ODS styles is using in-line styling.  In-line styling provides a convenient way to 
override the appearance of individual elements of your output without the need to use PROC TEMPLATE 
to create a complete style template.  This is done by specifying one or more pairs of attributes and values 
using the STYLE= option, which is available on specific statements in various procedures including the 
PRINT, REPORT, and TABULATE procedures, among others. 

Within PROC REPORT, in-line styling can be applied to several different report locations.  These 
locations are named COLUMN, HEADER, SUMMARY, REPORT, LINES, and CALLDEF.  Selecting one 
of these locations allows you to control which portions of the report output will be affected by the in-line 
style.  If a location is not specified, the in-line style will be applied to a default location.  The default 
location is different depending on which PROC REPORT statement contains the in-line style. 

Statement Default Location 

PROC REPORT REPORT 

DEFINE COLUMN and HEADER 

BREAK / RBREAK SUMMARY and LINES 

COMPUTE LINES 

 

The following PROC REPORT code shows how in-line styling can be applied to modify various attributes 
in various report locations.  The output is directed to an Excel spreadsheet using the ODS EXCEL 
destination, but the same PROC REPORT code could have been used with most other ODS destinations 
as well.  The resulting output is shown in Figure 17. 

ods excel file="c:\example.xlsx"; 

 

proc report data=sashelp.cars 

      style(report)=[cellpadding=3px]; 

  define make / style(column)=[color=red]; 

  define model/ style(header)=[color=green font_size=14pt]; 

  define type / style=[background_color=blue color=white]; 

run; 

 

ods excel close; 

Refer to Eslinger (2018) for a thorough explanation of customizing PROC REPORT output using ODS 
styles. 

Figure 17. Using In-Line Styling with PROC REPORT 

CONCLUSION 

The ODS EXCEL destination provides a powerful and flexible way to create Microsoft Excel workbooks 
directly from SAS.  The use of ODS styles and the various ODS EXCEL options provides for extensive 
customization of nearly every aspect of the resulting output.  This integration between SAS and Microsoft 
Excel allows one to have the best of both worlds – the power and functionality of the SAS system along 
with the flexibility and ubiquity of Microsoft Excel. 

